
1. Overview

This document lists TTCN changes needed to correct issues in the ATS ‘iwd-TTCN3-B2015-03_D16wk24.

3. Correction to EMM test case 9.3.1.6
Change 1 – Correction to function ‘f_TC_9_3_1_6_EUTRA’
	Function name
	f_TC_9_3_1_6_EUTRA ()

	Reason for change
	To allow performing a PDN Disconnect procedure for Multi-PDN UEs, the main behaviour sequence should refer at its end to steps 5 – 18b1 of the generic procedure 4.5.2.3 in TS 36.508. As a consequence also the UE state in which the TC body ends needs to be modified to “E1_IDLE”. 
(Note: This change requires a prose CR).

	Summary of change
	Extended at the end of fl_TC_9_3_1_6_Body() the call to generic procedure 4.5.2.3 from TS 36.508 and replaced UE state “E2_CONNECTED” with “E1_IDLE” when calling the Postamble function.

	TTCN module
	LTE_IRAT\9_3\NAS_ServiceRequest_UG.ttcn

	MCC160 Comment
	


Before change

...
  /*
   * @desc      REFERENCE TS 36.523-1 clause 9.3.1.6
   * @status    APPROVED (LTE_IRAT)
   */
  function f_TC_9_3_1_6_EUTRA() runs on EUTRA_PTC
  {
    f_EUTRA_NAS_InitIRATorEUTRAonly(LTE_AllCellsOnSamePLMN); // @sic R5-113805, R5-123550 sic@
    // create cell and send out system information
    f_EUTRA_CellConfig_Def(eutra_CellA);
    f_EUTRA_PreambleOtherRAT(eutra_CellA, -, true); // @sic R5s120264, R5s120575 sic@
    f_EUTRA_TestBody_Set(true);
    fl_TC_9_3_1_6_Body();
    // Postamble
    f_EUTRA_TestBody_Set(false);
    f_EUTRA_Postamble (eutra_CellA, E2_CONNECTED); // @sic R5s120264, , R5s120575 sic@
  } // end f_TC_9_3_1_6_EUTRA
  /*
   * @desc      test body of TC_9_3_1_6
   * @status    APPROVED (LTE_IRAT)
   */
  function fl_TC_9_3_1_6_Body() runs on EUTRA_PTC
  {
    var GutiParameters_Type v_GutiParams_A := f_EUTRA_CellInfo_GetGuti(eutra_CellA);
    var EUTRA_SecurityParams_Type v_SecurityParams := f_EUTRA_Security_Get();
    var NAS_MSG_Indication_Type v_NAS_Ind;
    var RRC_TransactionIdentifier v_RRC_TI := tsc_RRC_TI_Def;
    var NAS_AttDetValue_Type v_AttachType := f_GetEAttachType(NORMAL);
    var template AdditionalUpdateType v_AdditionalUpdateType := f_GetAdditionalUpdateType ();
    var float v_NoReactionTime;
    var charstring v_OtherPTC := f_EUTRA_GetOtherRAT();
    //+   The SS pages the UE using S-TMSI with CN domain indicator set to "PS".
    //@siclog "Step 1" siclog@
    f_EUTRA_UE_Page ( eutra_CellA,
                      cs_NasPaging_OneRecord(cs_PagingUE_Identity_S_TMSI(v_GutiParams_A.MME_Code,
                                                                         v_GutiParams_A.M_TMSI)));
    //+   The UE transmits a SERVICE REQUEST message. --> SERVICE REQUEST
    //@siclog "Step 2" siclog@
    v_NAS_Ind := f_EUTRA_RRC_ConnEst_DefWithNas(eutra_CellA,
                                                 v_RRC_TI,
                                                 cr_EstablishmentCause_mt_Access,
                                                 cr_NAS_Indication(tsc_SHT_IntegrityProtected,
                                                                   cr_508_SERVICE_REQUEST(v_SecurityParams.KSIasme)));
    //+   The SS transmits a SERVICE REJECT message with the EMM cause set to = "EPS services not allowed"
    //@siclog "Step 3" siclog@
    SRB.send ( cas_SRB1_NasPdu_REQ ( eutra_CellA,
                                       cs_TimingInfo_Now,
                                       cs_NAS_Request ( tsc_SHT_NoSecurityProtection,
                                                        cs_508_SERVICE_REJECT ( tsc_Cause_EPSServNotAllowed ) ) ) );
    //+   The SS releases the RRC connection.
    //@siclog "Step 4" siclog@
    f_EUTRA_RRC_ConnectionRelease(eutra_CellA);
    //+   Check: Does the UE transmit a SERVICE REQUEST message in the next 30 seconds?
    //@siclog "Step 5" siclog@
    v_NoReactionTime := f_EUTRA_SetTimerToleranceMax( eutra_CellA, nasTTCNTimer, 30.0 );
    if (f_EUTRA_RRC_RRCConnectionRequest_Check(eutra_CellA, v_NoReactionTime)) {
      f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step5");
    }
    //+  Check: Does the test result of the generic procedure in TS 36.508 subclause 6.4.2.5
    //          indicates that the UE responds to paging when paged with GUTI-1 and with CN
    //          domain indicator set to "PS"?
    //@siclog "Step 6" siclog@
    v_NoReactionTime := f_EUTRA_SetTimerToleranceMax( eutra_CellA, nonProtocolTimer, 3.0 );
    f_EUTRA_508Check_NASNoResponse ( eutra_CellA,
                                     cs_NasPaging_OneRecord ( cs_PagingUE_Identity_S_TMSI ( v_GutiParams_A.MME_Code, v_GutiParams_A.M_TMSI )),
                                     v_NoReactionTime );
    //@siclog "Step 7a1" siclog@
    if (v_OtherPTC != "NONE") {
      f_EUTRA_SetCellPower ( eutra_CellA, tsc_NonSuitableCellRS_EPRE );
      f_EUTRA_SendTriggerToOtherRAT(true);
      //@siclog "Step 9" siclog@
      f_EUTRA_SetCellPower ( eutra_CellA, tsc_ServingCellRS_EPRE );
    } else {
      //@siclog "Step 8a1" siclog@
      f_UT_SwitchOffUE (UT, true, CNF_REQUIRED); // @sic R5s1100395 sic@ @sic R5s120597r1 sic@
      //@siclog "Step 10" siclog@
      f_EUTRA_SwitchOnUEandStartIP(eutra_CellA, -, true); // @sic R5s120065 sic@ @sic R5s120597r1 sic@
    }
    //@siclog "Step 11" siclog@
    // AttachProcedure with specific message contents on step11
    v_NAS_Ind := f_EUTRA_RRC_ConnEst_DefWithNas(eutra_CellA,
                                                tsc_RRC_TI_Def,
                                                cr_EstablishmentCause_Any,
                                                cr_NAS_IndicationWithPiggybacking(tsc_SHT_NoSecurityProtection,
                                                                                  cdr_ATTACH_REQUEST_WithIMSIAndnoTAI(v_AttachType,
                                                                                                                      v_AdditionalUpdateType,
                                                                                                                      f_Imsi2MobileIdentity(px_IMSI_Def)),
                                                                                  cr_508_PDN_CONNECTIVITY_REQUEST(tsc_PdnRequest_InitialAttach,
                                                                                                                  f_GetPdnType())));
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 11");
    //@siclog "Step 12 - 18" siclog@
    f_EUTRA_InitialRegistration_Step5_16c1 (eutra_CellA, STATE2_IDLEUPDATE, NORMAL, v_NAS_Ind); // @sic R5s120264, R5s120575, R5s150830 sic@ @sic R5-160947 sic@
    // Test case finished now
    if (v_OtherPTC != "NONE") {
      f_EUTRA_SendTriggerToOtherRAT();
    }
 } // function fl_TC_9_3_1_6_Body
...

After change

...

  /*
   * @desc      REFERENCE TS 36.523-1 clause 9.3.1.6
   * @status    APPROVED (LTE_IRAT)
   */
  function f_TC_9_3_1_6_EUTRA() runs on EUTRA_PTC
  {
    f_EUTRA_NAS_InitIRATorEUTRAonly(LTE_AllCellsOnSamePLMN); // @sic R5-113805, R5-123550 sic@
    // create cell and send out system information
    f_EUTRA_CellConfig_Def(eutra_CellA);
    f_EUTRA_PreambleOtherRAT(eutra_CellA, -, true); // @sic R5s120264, R5s120575 sic@
    f_EUTRA_TestBody_Set(true);
    fl_TC_9_3_1_6_Body();
    // Postamble
    f_EUTRA_TestBody_Set(false);
    f_EUTRA_Postamble (eutra_CellA, E1_IDLE); // @sic R5s120264, , R5s120575 sic@
  } // end f_TC_9_3_1_6_EUTRA
  /*
   * @desc      test body of TC_9_3_1_6
   * @status    APPROVED (LTE_IRAT)
   */
  function fl_TC_9_3_1_6_Body() runs on EUTRA_PTC
  {
    var GutiParameters_Type v_GutiParams_A := f_EUTRA_CellInfo_GetGuti(eutra_CellA);
    var EUTRA_SecurityParams_Type v_SecurityParams := f_EUTRA_Security_Get();
    var NAS_MSG_Indication_Type v_NAS_Ind;
    var RRC_TransactionIdentifier v_RRC_TI := tsc_RRC_TI_Def;
    var NAS_AttDetValue_Type v_AttachType := f_GetEAttachType(NORMAL);
    var template AdditionalUpdateType v_AdditionalUpdateType := f_GetAdditionalUpdateType ();
    var float v_NoReactionTime;
    var charstring v_OtherPTC := f_EUTRA_GetOtherRAT();
    //+   The SS pages the UE using S-TMSI with CN domain indicator set to "PS".
    //@siclog "Step 1" siclog@
    f_EUTRA_UE_Page ( eutra_CellA,
                      cs_NasPaging_OneRecord(cs_PagingUE_Identity_S_TMSI(v_GutiParams_A.MME_Code,
                                                                         v_GutiParams_A.M_TMSI)));
    //+   The UE transmits a SERVICE REQUEST message. --> SERVICE REQUEST
    //@siclog "Step 2" siclog@
    v_NAS_Ind := f_EUTRA_RRC_ConnEst_DefWithNas(eutra_CellA,
                                                 v_RRC_TI,
                                                 cr_EstablishmentCause_mt_Access,
                                                 cr_NAS_Indication(tsc_SHT_IntegrityProtected,
                                                                   cr_508_SERVICE_REQUEST(v_SecurityParams.KSIasme)));
    //+   The SS transmits a SERVICE REJECT message with the EMM cause set to = "EPS services not allowed"
    //@siclog "Step 3" siclog@
    SRB.send ( cas_SRB1_NasPdu_REQ ( eutra_CellA,
                                       cs_TimingInfo_Now,
                                       cs_NAS_Request ( tsc_SHT_NoSecurityProtection,
                                                        cs_508_SERVICE_REJECT ( tsc_Cause_EPSServNotAllowed ) ) ) );
    //+   The SS releases the RRC connection.
    //@siclog "Step 4" siclog@
    f_EUTRA_RRC_ConnectionRelease(eutra_CellA);
    //+   Check: Does the UE transmit a SERVICE REQUEST message in the next 30 seconds?
    //@siclog "Step 5" siclog@
    v_NoReactionTime := f_EUTRA_SetTimerToleranceMax( eutra_CellA, nasTTCNTimer, 30.0 );
    if (f_EUTRA_RRC_RRCConnectionRequest_Check(eutra_CellA, v_NoReactionTime)) {
      f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step5");
    }
    //+  Check: Does the test result of the generic procedure in TS 36.508 subclause 6.4.2.5
    //          indicates that the UE responds to paging when paged with GUTI-1 and with CN
    //          domain indicator set to "PS"?
    //@siclog "Step 6" siclog@
    v_NoReactionTime := f_EUTRA_SetTimerToleranceMax( eutra_CellA, nonProtocolTimer, 3.0 );
    f_EUTRA_508Check_NASNoResponse ( eutra_CellA,
                                     cs_NasPaging_OneRecord ( cs_PagingUE_Identity_S_TMSI ( v_GutiParams_A.MME_Code, v_GutiParams_A.M_TMSI )),
                                     v_NoReactionTime );
    //@siclog "Step 7a1" siclog@
    if (v_OtherPTC != "NONE") {
      f_EUTRA_SetCellPower ( eutra_CellA, tsc_NonSuitableCellRS_EPRE );
      f_EUTRA_SendTriggerToOtherRAT(true);
      //@siclog "Step 9" siclog@
      f_EUTRA_SetCellPower ( eutra_CellA, tsc_ServingCellRS_EPRE );
    } else {
      //@siclog "Step 8a1" siclog@
      f_UT_SwitchOffUE (UT, true, CNF_REQUIRED); // @sic R5s1100395 sic@ @sic R5s120597r1 sic@
      //@siclog "Step 10" siclog@
      f_EUTRA_SwitchOnUEandStartIP(eutra_CellA, -, true); // @sic R5s120065 sic@ @sic R5s120597r1 sic@
    }
    //@siclog "Step 11" siclog@
    // AttachProcedure with specific message contents on step11
    v_NAS_Ind := f_EUTRA_RRC_ConnEst_DefWithNas(eutra_CellA,
                                                tsc_RRC_TI_Def,
                                                cr_EstablishmentCause_Any,
                                                cr_NAS_IndicationWithPiggybacking(tsc_SHT_NoSecurityProtection,
                                                                                  cdr_ATTACH_REQUEST_WithIMSIAndnoTAI(v_AttachType,
                                                                                                                      v_AdditionalUpdateType,
                                                                                                                      f_Imsi2MobileIdentity(px_IMSI_Def)),
                                                                                  cr_508_PDN_CONNECTIVITY_REQUEST(tsc_PdnRequest_InitialAttach,
                                                                                                                  f_GetPdnType())));
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 11");
    //@siclog "Step 12 - 18" siclog@
    f_EUTRA_InitialRegistration_Step5_18b1(eutra_CellA, STATE2_IDLEUPDATE, NORMAL, v_NAS_Ind); // @sic R5s120264, R5s120575, R5s150830 sic@ @sic R5-160947 sic@
    // Test case finished now
    if (v_OtherPTC != "NONE") {
      f_EUTRA_SendTriggerToOtherRAT();
    }
 } // function fl_TC_9_3_1_6_Body
... 
